# Why does Simplify work diferently for (a+b)/e versus (a+b+c)/e?

I am playing around with Simplify amd assumptions. I noticed that:

Simplify[x > (a + b)/e, x < (a + b)/e]

False


But if I add more variables to the numerator, Simplify is not able to get False

Simplify[x > (a + b + c)/e, x < (a + b + c)/e]

x > (a + b + c)/e


Why is there such a difference and how can I avoid the latter situation?

• Note that Reduce[{x > (a + b + c)/e, x < (a + b + c)/e}] does return False as expected, so that might be a work-around. – Michael Seifert Jul 29 '20 at 14:38
• Simplify[x > (a + b + c)/e && x < (a + b + c)/e] – Alan Jul 29 '20 at 14:49
• – Michael E2 Jul 29 '20 at 16:16

You can set a higher value for the system sub-option "AssumptionsMaxNonlinearVariables" than its default value 4:

"AssumptionsMaxNonlinearVariables" /.
"SimplificationOptions" /.
SystemOptions["SimplificationOptions"]

 4

Simplify[x > (a + b + c)/e, x < (a + b + c)/e]

 x > (a + b + c)/e

Simplify[x > (a + b + c + z)/e, x < (a + b + c + z)/e]

 x > (a + b + c + z)/e

ClearSystemCache[]
SetSystemOptions["SimplificationOptions" -> {"AssumptionsMaxNonlinearVariables" -> 7}];

Simplify[x > (a + b + c)/e, x < (a + b + c)/e]

False

Simplify[x > (a + b + c + z)/e, x < (a + b + c + z)/e]

False